
1. Εισαγωγή 
 

 

Το λογισμικό που αναπτύχθηκε από την Κύρια Ερευνητική Ομάδα του Εθνικού Μετσοβίου 
Πολυτεχνείου, αφορά στις μεθοδολογίες ανάλυσης ευστάθειας σε πραγματικά προβλήματα 
πολυστοχικού προγραμματισμού (πολυκριτηριακός μαθηματικός προγραμματισμός). Αυτές 
οι μεθοδολογίες ανάλυσης ευστάθειας αναπτύχθηκαν κατά τη διάρκεια της Δράσης Δ.10 και 
αξιολογήθηκαν πειραματικά κατά την Τεχνική Έκθεση Π.11. Τέλος, χρησιμοποιήθηκαν σε 
πραγματικά προβλήματα πολυστοχικού προγραμματισμού μεγάλης κλίμακας κατά την 
υλοποίηση της Δράσης Δ.12.  

Στόχος της παρούσας Τεχνικής Έκθεσης είναι η συνοπτική παρουσίαση των μέτρων 
αξιολόγησης της ευστάθειας στα προβλήματα πολυστοχικού προγραμματισμού και κατόπιν 
η παρουσίαση του σχετικού λογισμικού που αναπτύχθηκε. Παράλληλα, με την παρουσίαση 
του λογισμικού και την παράθεση όλου του πηγαίου κώδικα στο Παράρτημα του παρόντος 
Παραδοτέου, θα δοθούν και οδηγίες σωστής χρήσης του με τη μορφή εγχειριδίου.  

Ο κώδικας του αναπτυχθέντος λογισμικού/αλγορίθμων είναι γραμμένος σε γλώσσα 
μοντελοποίησης, και πιο συγκεκριμένα στην πλατφόρμα GAMS (General Algebraic Modeling 
System). Η πλατφόρμα GAMS αποτελεί ένα πολύ ισχυρό εργαλείο που εξειδικεύεται σε 
προβλήματα μαθηματικής βελτιστοποίησης όλων των ειδών (Linear Programming, Integer 
Programming, Non Linear Programming, Mixed Complementarity Problems, Quadratically 
Constrained Programs, κλπ.). Η πλατφόρμα ενσωματώνει όλους τους αντίστοιχους solvers 
που απαιτούνται για την επίλυση των συγκεκριμένων ειδών προβλημάτων βελτιστοποίησης. 

Το λογισμικό GAMS μπορεί κανείς να το κατεβάσει δωρεάν από την ιστοσελίδα 
http://www.gams.com/ και να κάνει απεριόριστη χρήση του, με τον περιορισμό ότι τα 
προβλήματα που θα επιλύει δε θα ξεπερνούν σε μέγεθος τις 300 μεταβλητές και 
περιορισμούς. Για την επίλυση μεγαλύτερων προβλημάτων βελτιστοποίησης, είναι αναγκαία 
η αγορά του, είτε με business ή με ακαδημαϊκούς όρους.  

Σημειώνεται ότι η πλατφόρμα GAMS αποτελεί μια εκ των υψηλά εξειδικευμένων σε 
προβλήματα βελτιστοποίησης, και το γεγονός ότι εστιάζει μόνο στην αντιμετώπιση τέτοιων 
προβλημάτων, την κάνει μία εκ των ισχυρότερων, σύμφωνα με τους εμπειρογνώμονες. 
Παράλληλα, η εκτεταμένη βιβλιογραφία (manuals, documentations, tutorials, model 
libraries, κ.α.), που έχει αναπτυχθεί, βοηθούν εξαιρετικά ένα νέο χρήστη της. 

  

http://www.gams.com/

2. Εγχειρίδιο χρήσης πολυστοχικού λογισμικού 
 

 

Το λογισμικό που αναπτύχθηκε από το ΕΜΠ, είναι ενσωματομένο σε τρία διαφορετικά 
αρχεία GAMS. Αυτά είναι τα: 

 MOIP – MultiObjective Integer Programming, 

 MOLP - MultiObjective Linear Programming, και 

 MOMIP – MultiObjective Mixed Integer Programing. 

Τα τρία διαφορετικά αρχεία αφορούν στην ανάλυση της ευστάθειας στα τρία πιο βασικά 
προβλήματα πολυστοχικού μαθηματικού προγραμματισμού και θα εξεταστούν αναλυτικά 
παρακάτω. Τα πρώτο αρχείο λογισμικού (MOIP) αφορά στην επιλογή βέλτιστου 
χαρτοφυλακίου έργων, ενώ τα άλλα δύο (MOLP και MOMIP) στην βελτιστοποίηση 
χαρτοφυλακίου μετοχών. Και στα τρία δίνεται παρατείθεται μια πραγματική εφαρμογή, 
όπως παρουσιάστηκαν στις Τεχνικές Εκθέσεις Π.11 και Π.12. 

Το πρώτο βήμα που πρέπει να κάνει ο χρήστης είναι η λήψη και εγκατάσταση της 
πλατφόρμας GAMS από το site της εταιρείας: http://www.gams.com/download/. Κατόπιν, 
κάνοντας διπλό κλικ σε ένα εκ των τριών αρχείων, αυτό θα ανοίξει και θα είναι έτοιμο προς 
άμεση επεξεργασία. Διαφορετικά, ο χρήστης μπορείς να ανοίξει το πρόγραμμα GAMS, να 
πατήσει “File” και κατόπιν “Open” και να κάνει browse το αρχείο, ώστε να το ανοίξει. 

 

2.1 MOIP – Multi-Objective Integer Programming 

Το συγκεκριμένο αρχείο εστιάζει στην επίλυση ενός προβλήματος βελτιστοποίησης 
πολυστοχικού ακέραιου μαθηματικού προγραμματισμού, και πιο συγκεκριμένα, στην 
επιλογή βέλτιστου χαρτοφυλακίου έργων. Παρέχει το παράδειγμα που αναφέρεται στις 
δημοσιευμένες επιστημονικές εργασίες  Mavrotas et al. (2015a) και Mavrotas et al. (2015b).  

Στην αρχή του κώδικα, ο χρήστης καλείται να ορίσει έξι διαφορετικά set παραμέτρων. 
Συγκεκριμένα, τα διαφορετικά projects που αξιολογούνται (i), τη γεωγραφική τους θέση 
(geo), το είδος τεχνολογίας που αφορούν (Tech) και τις αντικειμενικές συναρτήσεις (k). Τα 
άλλα δύο sets είσάγονται με τη μορφή πίνακα και είναι: η αντιστοίχηση έργων σε 
γεωγραφική περιοχή (geo, i) και η αντιστοίχηση των έργων σε τεχνολογίες (tech, i). Τα set, 
αφορούν στην ουσία τους δείκτες που διαφοροποιούν τις παραμέτρους μεταξύ τους, και 
απαιτούνται σε κάθε αρχείο GAMS. Τα έξι διαφορετικά sets, εμφανίζονται με μπλε χρώμα 
στο Σχήμα 2.1 και οι τιμές που παίρνουν με πράσινο. Τα χρωματισμένα με γκρι χρώμα είναι 
σχόλια και όχι εντολές που διαβάζονται από τον solver. 

Μετά τον ορισμό όλων των sets, εισάγονται οι παράμετροι του μοντέλου (parameters) Και 
κάποιες σταθερές (scalars) που αφορούν σε ποσότητες του μοντέλου οι οποίες είναι γνωστές, 
π.χ. το budg(i) που δηλώνει το συνολικό προϋπολογισμό κάθε έργου. Οι παράμετροι όπως 
και τα set, μπορούν να εισαχθούν ή να διαβαστούν κατευθείαν από αρχεία .txt, xls ή .csv, 
όταν ο όγκος των δεδομένων είναι πολύ μεγάλος. Ο χρήστης παραπέμπεται στο εκτενές και 

http://www.gams.com/download/

εξαιρετικό tutorial του GAMS, για την εύρεση των κατάλληλων σχετικών εντολών (Rosenthal. 
2010). 

 

 

Σχήμα 2.1: Στιγμιότυπο οθόνης κατά την αρχή του πηγαίου κώδικα MOIP 

 

Μετά την εισαγωγή των παραμέτρων, σειρά να εισαχθούν έχουν οι άγνωστες ποσότητες του 
μοντέλου, δηλαδή οι μεταβλητές του (variables), οι οποίες διαχωρίζονται σε θετικές, 
αρνητικές, πραγματικές, ακέραιες, δυαδικές, κλπ. Στο συγκεκριμένο μοντέλο, οι μεταβλητές 
που ενδιαφέρουν τον αναλυτή κατά κύριο λόγο είναι οι μεταβλητές X(i), που αποτελούν και 
τις μεταβλητές απόφασης της βελτιστοποίησης. Πρόκειται για δυαδικές μεταβλητές (0,1) που 

http://www.gams.com/docs/document.htm
http://www.gams.com/docs/document.htm

ελέγχουν και ορίζουν αν το i project, θα χρηματοδοτηθεί ή όχι. Στην πρώτη περίπτωση 
παίρνουν την τιμή 1, στη δεύτερη την τιμή 0.  

Στη συνέχεια, ορίζονται οι εξισώσεις/ανισώσεις που ενσωματώνουν τις μεταβλητές και τις 
παραμέτρους, που στην ουσία περιγράφουν τους περιορισμούς του προβλήματος, και τέλος 
δίνεται η αντικειμενική συνάρτηση προς βελτιστοποίηση. Στο παρόν παράδειγμα έχουν 
επιλεχθεί 5 αντικειμενικές συναρτήσεις, όσες δηλαδή είχαν οριστεί από το set “k”. Αυτές 
είναι, η περιφερειακή ανάπτυξη, η ανεργία, η μείωση εκπομπών διοξειδίου του άνθρακα, η 
χρήση γης και ο οικονομικός δείκτης IRR (συντελεστής εσωτερικής απόδοσης). 

 

 

Σχήμα 2.2: Στιγμιότυπο οθόνης των μεταβλητών και περιορισμών του κώδικα ΜΟIP 

 

H συνέχεια του κώδικα αφορά στην ανάλυση της ευστάθειας, μέσω πολλαπλών τρεξιμάτων 
και επιλύσεων του μοντέλου, κάνοντας χρήση του αλγορίθμου Monte Carlo και της 
πολυστοχικής μεθόδου Augmented Weighted Tchebyscheff (βλ. Mavrotas et al. 2015b). 

Εκεί παράγεται ο Πίνακας Κερδών 5Χ5 των αντικειμενικών συναρτήσεων και δηλώνονται οι 
παράμετροι η τιμή των οποίων είναι αβέβαιη, αλλά ακολουθεί μια συνάρτηση πυκνότητας 

πιθανότητας. Το μοντέλο τρέχει looped όσες φορές δηλωθεί από το χρήστη, και έτσι 
αποκτώνται στατιστικά αποτελέσματα των αποτελεσμάτων. Αυτή η ανάλυση δείχνει ποιες 
λύσεις και ποια έργα εμφανίζονται συχνότερα στο βέλτιστο χαρτοφυλάκιο έργων, πράγμα 
που σημαίνει ότι είναι και πιο ευσταθείς.  

Ο αλγόριθμος σε όλη του την έκταση παρέχει πλήρη σχολιασμό και οδηγίες για τη σωστή και 
αποτελεσματική του χρήση, χωρίς σφάλματα. 

 

2.2 MOLP – Multi-Objective Linear Programming  

Το συγκεκριμένο αρχείο εστιάζει στην επίλυση ενός προβλήματος βελτιστοποίησης 
πολυστοχικού γραμμικού προγραμματισμού, και πιο συγκεκριμένα, στην επιλογή βέλτιστου 
χαρτοφυλακίου μετοχών. Παρέχει το παράδειγμα που αναφέρεται στην Τεχνική Έκθεση Π.11 
του ΘΑΛΗ. 

Στην αρχή του κώδικα, ο χρήστης καλείται να ορίσει εννέα διαφορετικά set παραμέτρων. 
Συγκεκριμένα, τα διαφορετικά κριτήρια/αντικειμενικές συναρτήσεις (Κ), τις παρατηρήσεις 
(Τ), τις μετοχές υπό αξιολόγηση (Ι), τους εκδότες (IS), τις αγορές (MRK), που στο συγκεκριμένο 
παράδειγμα είναι μόνο μία, τους τομείς δραστηριότητας (SCT), καθώς και η χαρτογράφηση 
εκδοτών και μετοχών (IS, I), η αντιστοίχιση τομέων και μετοχών (SCT, I) και η χαρτογράφηση 
μετοχών και αγορών (MRK, I). Τα set εντός της πλατφόρμας GAMS απεικονίζονται στο Σχήμα 
2.3, μαζί με τον αρχικό σχολιασμό του κώδικα. 

Κατόπιν, εισάγεται μια πληθώρα παραμέτρων και σταθερών, όπως η πορεία των μετοχών τα 
τελευταία δύο χρόνια, κάποια άνω και κάτω όρια των αντικειμενικών συναρτήσεων, τα τέλη 
αγορών και πωλήσεων, κ.α. Πολλές παράμετροι όπως φαίνεται, δεν εισάγονται σε νουμερική 
μορφή, αλλά καλείται μέσω της εντολής “$include” το αντίστοιχο αρχείο GAMS στο οποίο 
αυτές έχουν οριστεί.  

Στη συνέχεια εισάγονται κατά τα γνωστά οι μεταβλητές του μοντέλου, 
συμπεριλαμβανομένων και των μεταβλητών απόφασης που είναι το ποσοστό κάθε μετοχής 
στο βέλτιστο χαρτοφυλάκιο (W(I)). Τέλος, εισάγονται οι περιορισμοί υπό τη μορφή 
εξισώσεων και ανισοτικών εξισώσεων, και η αντικειμενικές συναρτήσεις της μεγιστοποίησης 
του κέρδους και ελαχιστοποίησης του ρίσκου. 

Το μοντέλο επιλύεται με τη χρήση της πολυστοχικής μεθόδου ε-constraint, και συγκεκριμένα 
με την βελτιωμένη της έκδοση την AUGMECON. Η ανάλυση της ευστάθειας των λύσεων 
πραγματοποιείται πάλι με τη χρήση του αλγορίθμου Monte Carlo και των πολλαπλών 
δειγματοληψιών και επιλύσεων του μοντέλου, για την εξαγωγή στατιστικών αποτελεσμάτων. 

Στο Σχήμα 2.4 παρουσιάζεται η επίλυση της πολυστοχικής μεθόδου Tchebytcheff, καθώς και 
ο καθορισμός  συναρτήσεων πυκνότητας πιθανότητας για τα βάρη των αντικειμενικών 
συναρτήσεων. Παρουσιάζεται επίσης, ο αλγόριθμος Monte Carlo, μέσω του βρόγχου 
πολλαπλών επιλύσεων του μοντέλου (loop).  

Τελικά, τα αποτελέσματα που εξάγονται αποθηκεύονται σε αρχείο .txt στο φάκελο που ορίζει 
ο χρήστης εντός του πηγαίου κώδικα.  

Σχήμα 2.3: Στιγμιότυπο οθόνης κατά την αρχή του πηγαίου κώδικα MOLP. 

2.3 MOΜΙP – Multi-Objective Mixed Integer Programming  

Το αρχείο κώδικα MOMIP είναι κατά πολύ μεγάλο ποσοστό αντίστοιχο του MOLP που μόλις 
περιγράφηκε, με τη διαφορά ότι αυτό είναι ελαφρώς πιο εξελιγμένο, καθώς περιέχει και 
κάποιες επιπρόσθετες ακέραιες μεταβλητές, όπως την δυαδική μεταβλητή Β(i), που δηλώνει 
την ύπαρξη μιας μετοχής i στο βέλτιστο χαρτοφυλάκιο. Ο συγκεκριμένος αλγοριθμικός 
κώδικας απαιτεί σαφώς μεγαλύτερη υπολογιστική ισχύ από τον προηγούμενο καθώς 
πρόκεται για Combinatorial Optimization πρόβλημα (βλ. Mavrotas et al. 2015a). 

 

 

Σχήμα 2.4: Στιγμιότυπο οθόνης της μεθόδου Tchebytcheff στο MOLP μοντέλο  

Παράρτημα 
 

 

$ontext 
This is a model for the multiobjective portfolio optimization 
 
In this example we consider an already existent porfolio 
the weights in the existent portfolio are given in parameters w0 
 
We have: 
3 objective functions (criteria) which are (1) return, (2) MAD and (3) Liquidity 
50 stocks 
104 weeks for past data (2 years) 
1 market (EUR) 
10 sectors 
50 issuers 
first 15 stocks as high capitalization 
 
Constraints 
Portfolio's beta = 0.7-1.3 
The share of the participating in the portfolio stocks is between 0.03 and 0.1 
The number of stocks in the porfolio is between 10 and 20 
5-10-40 rule 
 
$offtext 
 
$inlinecom [ ] 
$eolcom // 
***************************************************************************
*** 
$STitle Example model definitions 
 
SETS 
K  criteria /1*3/ 
*DK descriptive attributes (not used as objective functions)  /1*1/ 
T  observations  /1*104/ 
I  stocks /1*50/ 
*NZW(I) stocks with nonzero weight in existent portfolio 
/2,3,6,9,11,14,15,16,17,18,20,21,24,26,27,28,29,32,34,37,39,40,43,44,46/ 
 
IS issuers /1*50/ 
ISI(IS,I) mapping of issuers to stocks 
/1.1, 2.2, 3.3, 4.4, 5.5, 6.6, 7.7, 8.8, 9.9, 10.10, 
11.11, 12.12, 13.13, 14.14, 15.15, 16.16, 17.17, 18.18, 19.19, 20.20, 



Δ14 - Ανάπτυξη αλγορίθμου ευσταθούς 
πολυκριτήριας βελτιστοποίησης 

Π14.1 – Τεχνική Έκθεση 
Π14.2 – Υπολογιστική Υλοποίηση 

 

 

ΘΑΛΗΣ – Πανεπιστήμιο Πειραιά – Μεθοδολογικές προσεγγίσεις για τη μελέτη της 
ευστάθειας σε προβλήματα λήψης αποφάσεων με πολλαπλά κριτήρια 

Σελίδα 13 από 22 

 

21.21, 22.22, 23.23, 24.24, 25.25, 26.26, 27.27, 28.28, 29.29, 30.30, 
31.31, 32.32, 33.33, 34.34, 35.35, 36.36, 37.37, 38.38, 39.39, 40.40, 
41.41, 42.42, 43.43, 44.44, 45.45, 46.46, 47.47, 48.48, 49.49, 50.50/ 
 
*markets 
MRK markets /EUR/ 
MRKI(MRK,I) mapping of markets to stocks EUR 1-50 
/EUR.1*50/ 
 
*super sectors 
SCT sectors /1*10/ 
SCTI(SCT,I) mapping of sectors to stocks 
/ 
1.(1,5,33), 
2.(2,31,39,41,42,48,49), 
3.(3,16,21,38,43,46), 
4.(4,10,18,23,24,27,28,29,30,34,36,40,45,47,50), 
5.(13,25), 
6.(6,7,8,15,32,44), 
7.(9,19,20,26,37), 
8.(12), 
9.(11,17,22,35), 
10.(14) 
/ 
 
*1        Energy 
*2        Industrial 
*3        Communications 
*4        Financial 
*5        Consumer|Cyclical 
*6        Consumer|Non-cyclical 
*7        Utilities 
*8        Diversified 
*9        Basic Materials 
*10        Technology 
 
*blue chips 
BLCHP(I) stocks that are considered blue chips /1*15/ 
; 
 
TABLE dr(I,T) the T-th weekly deviation from 2-year mean of the I-th stock 
$include "c:\gams\eurostoxx50\dr_50_104_1.txt" ; 
; 
Parameter r(I) the 2- year mean of the return of the stocks 
/ 
$include "c:\gams\eurostoxx50\ret_50_1.txt" ; 
/ ; 
Parameter liq(I) the 2- year mean of the liquidity of the stocks in million euros 

/ 
$include "c:\gams\eurostoxx50\liq_50_1.txt" ; 
/ ; 
Parameter beta(I) the 2- year beta of the stocks 
/ 
$include "c:\gams\eurostoxx50\beta_50_1.txt" ; 
/ ; 
Parameter divy(I) the 2- year beta of the stocks 
/ 
$include "c:\gams\eurostoxx50\divy_50_1.txt" ; 
/ ; 
Parameter w0(I) the weights in the existent portfolio 
/ 
$include "c:\gams\eurostoxx50\initial_50_1.txt" ; 
/ ; 
 
Parameter w0fixwis(IS) the weights of the non-equities of each issuer that remains constant 
* we need it just for the 5-10-40 rule application that holds for all investments 
* of the same issuer, not only for stocks 
*/ 
*$include "c:\gams\x0fixwis_167.txt" ; 
*/ ; 
loop (I, w0fixwis(IS)=0); 
 
Parameter dir(K) direction of the objective functions 
/ 
1  1 
2  -1 
3  1 
/ ; 
 
Parameter ubobj(K) upper bound for the objective functions 
/ 
1  INF 
2  INF 
3  INF 
/ ; 
 
Parameter lbobj(K) lower bound for the objective functions 
/ 
1  -inf 
2  -INF 
3  -INF 
/ ; 
 
Parameter bcom(MRK) buying commission 
/ 
*ATH  0.5 

EUR   0.1 
*LON  0.4 
/ 
 
Parameter scom(MRK) selling commission 
/ 
*ATH  0.8 
EUR   0.15 
*LON  0.6 
/ 
 
Parameter mrktup(MRK) upper bound for markets 
/ 
*ATH  0.75 
EUR   1 
*LON  0.25 
/ 
 
Parameter mrktlo(MRK) lower bound for markets 
/ 
*ATH  0.25 
EUR   0 
*LON  0.1 
/ 
 
Parameter sectup(SCT) upper bound for sectors 
/ 
1   INF 
2   INF 
3   INF 
4   INF 
5   INF 
6   INF 
7   INF 
8   INF 
9   INF 
10  INF 
/ 
 
Parameter sectlo(SCT) lower bound for sectors 
/ 
1    0 
2    0 
3    0 
4    0 
5    0 
6    0 
7    0 

8    0 
9    0 
10   0 
/ 
 
; 
scalar 
lb lower bound for share in portfolio /0.03/ 
ub upper bound for share in portfolio /0.1/ 
minstocks lower bound for stocks in the portfolio /10/ 
maxstocks upper bound for stocks in the portfolio /20/ 
bclb lower bound for blue chips  /0.0/ 
bcub upper bound for blue chips  /1.0/ 
numstock number of stocks in the portfolio 
earlyex  number of early exits from the loops 
elapsed_time elapsed time for payoff and e-sonstraint 
pareto_num  number of Pareto optimal solutions 
start start time 
finish finish time 
numloop number of loops in e-constraint 
betalb   lower bound for portfolio'w beta /0.7/ 
betaub   upper bound for portfolio's beta /1.3/ 
divylb   lower bound for portfolio's dividend's yield /0/ 
d_abs         allowable increase or decrease from current weight in portfolio (absolute) /0.1/ 
d_rel         allowable increase or decrease from current weight in portfolio (relative) /1.0/ 
transact  reduction of return due to transaction costs 
; 
 
POSITIVE VARIABLES 
W(I)  the share of stock I in the portfolio 
Y(T)  auxiliary variable for the MAD definition according to Konno and Yamazaki 1991 
WB(I) the share of stock NZJ that is bought 
WS(I) the share of stock NZJ that is sold 
OV5W(IS) The share of stock I over 5% 
*AUX(IS) The share of stock I     //for stein's modelling 
PORTFBETA  portfolio's beta 
PORTFLIQ    portfolio's liquidity 
 
 free variables 
Z(K)  value of the objective functions 
*DZ(DK) descriptive attributes (not objective functions) 
; 
 
 
BINARY VARIABLES 
B(I)  the existence of stock I in the portfolio 
OV5B(IS) indicating if the stock J is over 5% 
; 

EQUATIONS 
*DECLARATION OF EQUATIONS 
*definition of objective functions 
EQ_OBJ1         objective function 1  (return) 
EQ_OBJ2         objective function 2  (mad) 
EQ_OBJ3         objective function 3  (liquidity) 
EQ_BETA         calculation of portfolio's beta 
EQ_LIQ         calculation of portfolio's liquidity 
 
*General constraints 
EQ_LO(I)        lower bound for share in portfolio for stock I 
EQ_UP(I)        upper bound for share in portfolio for stock I 
EQ_BLCHPLO      lower bound for share in portfolio for blue chips 
EQ_BLCHPUP      upper bound for share in portfolio for blue chips 
EQ_SUMBJLO      lower bound for number of stocks in the portfolio 
EQ_SUMBJUP      upper bound for number of stocks in the portfolio 
*EQ_DEV_RELB(NZW)     relative deviation from current portfolio shares - upper 
*EQ_DEV_RELS(NZW)     relative deviation from current portfolio shares - lower 
 
*specific constraints 
*EQ_MRKTLO(MRK)  lower bound for share of Market's 1 stocks 
*EQ_MRKTUP(MRK)  upper bound for share of Market's 1 stocks 
*EQ_SECTLO(SCT)  lower bound for share of Sector's 3 stocks 
*EQ_SECTUP(SCT)  upper bound for share of Sector's 3 stocks 
 
*auxiliary constraints 
EQ_SUMXJ        sum of shares equals 1 
EQ_BAL(I)       balance for stocks 
EQ_MAD1(T)      auxiliary equation for the definition of MAD 
EQ_MAD2(T)      auxiliary equation for the definition of MAD 
 
* 5-10-40 rule constraints 
EQ_OV5_1(IS)     auxiliary constraint 1 
EQ_OV5_2(IS)     auxiliary constraint 2 
*EQ_OV5_3(IS)     auxiliary constraint 2 
EQ_LINK_OV5(IS)  auxiliary constraint 3 
EQ_CONSTR_40     auxiliary constraint 4 
; 
 
*DESCRIPTION OF EQUATIONS 
*definition of objective functionsand descriptive criteria 
EQ_OBJ1..    sum(I,W(I)*r(I)) 
             - SUM(MRK,SUM(MRKI(MRK,I),(WB(I)*bcom(MRK)+WS(I)*scom(MRK))))=e=Z('1'); 
 
EQ_OBJ2..    sum(T,Y(T))/card(T)=e=Z('2'); 
EQ_OBJ3..    sum(I,W(I)*divy(I)) =e= Z('3'); 
EQ_BETA..    sum(I,W(I)*beta(I)) =e= PORTFBETA; 

EQ_LIQ..    sum(I,W(I)*liq(I)) =e= PORTFLIQ; 
*EQ_EARNY..    sum(I,W(I)*divy(I)) =e= PORTFEY; 
 
*General constraints 
EQ_LO(I)..   W(I) - lb*B(I) =g= 0 ; 
EQ_UP(I)..   W(I) - ub*B(I) =l= 0 ; 
EQ_BLCHPLO.. sum(BLCHP,W(BLCHP))=g= bclb ; 
EQ_BLCHPUP.. sum(BLCHP,W(BLCHP))=l= bcub ; 
EQ_SUMBJLO.. sum(I,B(I))=g= minstocks ; 
EQ_SUMBJUP.. sum(I,B(I))=l= maxstocks ; 
*EQ_DEV_RELB(NZW).. WB(NZW) =l= d_rel*w0(NZW); 
*EQ_DEV_RELS(NZW).. WS(NZW) =l= d_rel*w0(NZW); 
 
*specific constraints for markets 
*EQ_MRKTLO(MRK).. sum(MRKI(MRK,I),W(I)) =g= mrktlo(MRK) ; 
*EQ_MRKTUP(MRK).. sum(MRKI(MRK,I),W(I)) =l= mrktup(MRK) ; 
 
*specific constraints for sectors 
*EQ_SECTLO(SCT).. sum(SCTI(SCT,I),W(I)) =g= sectlo(SCT) ; 
*EQ_SECTUP(SCT).. sum(SCTI(SCT,I),W(I)) =l= sectup(SCT) ; 
 
*auxiliary constraints 
EQ_SUMXJ..   sum(I, W(I))=e= 1-sum(IS,w0fixwis(IS)) ; 
EQ_BAL(I)..  W(I)-WB(I)+WS(I)=e=w0(I); 
EQ_MAD1(T).. sum(I,W(I)*dr(I,T))+ Y(T) =g= 0; 
EQ_MAD2(T).. sum(I,W(I)*dr(I,T))- Y(T) =l= 0; 
 
*5-10-40 rule constraints 
*$ontext 
EQ_OV5_1(IS).. sum(ISI(IS,I),W(I))+w0fixwis(IS) - 0.05* OV5B(IS) =l= 0.05; 
EQ_OV5_2(IS).. OV5W(IS) - 0.05*OV5B(IS) =L= 0; 
EQ_LINK_OV5(IS).. sum(ISI(IS,I),W(I))+w0fixwis(IS)-0.05 =L= OV5W(IS); 
EQ_CONSTR_40.. SUM(IS,OV5W(IS))+0.05*SUM(IS,OV5B(IS)) =l= 0.4; 
*$offtext 
 
$ontext 
*STEIN'S MODELLING 
EQ_OV5_1(IS).. sum(ISI(IS,I),W(I))+w0fixwis(IS) - 0.05* OV5B(IS) =l= 0.05; 
EQ_OV5_2(IS).. AUX(IS) - 0.1*OV5B(IS) =L= 0; 
EQ_OV5_3(IS).. AUX(IS) - sum(ISI(IS,I),W(I))-w0fixwis(IS) =L= 0; 
EQ_LINK_OV5(IS).. sum(ISI(IS,I),W(I)) +w0fixwis(IS)+ 0.1 *OV5B(IS) -AUX(IS) =L= 0.1; 
EQ_CONSTR_40.. SUM(IS,AUX(IS)) =l= 0.4; 
$offtext 
 
*bounds on objective functions 
Z.up(K)=ubobj(K); 
Z.lo(K)=lbobj(K); 
*bounds on other attributes 

PORTFBETA.UP=betaub; 
PORTFBETA.LO=betalb; 
*PORTFDY.LO=divylb; 
 
*bounds due to maximum absolute deviation from current portfolio 
*WB.up(NZW)=d_abs; 
*WS.up(NZW)=d_abs; 
 
model eur_50 /all/ ; 
 
FILE fx /c:\gams\eur_50_results_tch.txt/ ; 
fx.pw=2000; 
put fx ; 
 
start=jnow; 
 
*--------------------------------------------------------------------------- 
$STitle payoff matrix 
 
Set kk(k)     active objective function in constraint allobj 
 
Parameter 
payoff(k,k)  payoff tables entries 
maxobj(k)  maximum value from the payoff table 
minobj(k)  minimum value from the payoff table 
 
Variables 
   obj        auxiliary variable during the construction of the payoff table 
Equations 
   allobj     all the objective functions in one expression; 
 
allobj..  sum(kk, dir(kk)*z(kk)) =e= obj; 
 
Model mod_payoff    / eur_50, allobj / ; 
 
Alias(k,kp); 
 
option optcr=0.0; 
option limrow=0, limcol=0, solprint=off ; 
 
* Generate payoff table applying lexicographic optimization 
loop(kp, 
  kk(kp)=yes; 
  repeat 
    solve mod_payoff using mip maximizing obj; 
    payoff(kp,kk) = z.l(kk); 
    z.fx(kk) = z.l(kk); // freeze the value of the last objective optimized 
    kk(k++1) = kk(k);   // cycle through the objective functions 

  until kk(kp); kk(kp) = no; 
* release the fixed values of the objective functions for the new iteration 
  z.up(k) = inf; z.lo(k) =-inf; 
); 
if (mod_payoff.modelstat<>1 and mod_payoff.modelstat<>8, abort 'no optimal solution for 
mod_payoff'); 
 
PUT fx ' PAYOFF TABLE'/   ; 
loop (kp, 
        loop(k, put payoff(kp,k):12:2); 
        put /; 
     ); 
put fx /; 
 
*display payoff; 
minobj(k)=smin(kp,payoff(kp,k)); 
maxobj(k)=smax(kp,payoff(kp,k)); 
 
*--------------------------------------------------------------------------- 
$STitle Tchebyscheff model 
scalar 
elapsed_time elapsed time for payoff and e-constraint 
start start time 
finish finish time 
iter  counter for iterations 
*r auxiliary parameter 
MCiter number of Monte Carlo iterations /1000/ 
okflag flag to indicate if we sample weight coefficients correctly 
totiter  total number of MC iterations 
; 
 
 
Parameter 
   zideal(k)  z** of Tchebysheff method 
   zdefn(k)   definition point z with dash over 
*   epsilon(k) parameter for creation of z** Steuer p. 422 (1%-10% of z) 
   rho        parameter in the objective function Steuer p. 429 (0.01-0.0001) /0.001/ 
 
* initial weights 0.1, 0.2, 0.2, 0.1, 0.4 
parameter wmin(K) weight coefficent intervals - left edge 
/ 
1        0.18 
2        0.45 
3        0.27 
/ 
parameter wmax(K) weight coefficent intervals - right edge 
/ 
1        0.22 

2        0.55 
3        0.33 
/ 
parameter wk(K)  weight coefficient; 
 
loop(k, 
    if (dir(k)=1, 
       zideal(k)=maxobj(k)+1; 
       zdefn(k)=minobj(k); 
    else 
       zideal(k)=minobj(k)-1; 
       zdefn(k)=maxobj(k); 
       ); 
     ); 
 
Variables 
   ALPHA      variable of tchebychev method 
   TCH_OBJ    objective function of augmented Tchebysheff 
Equations 
   tch_objfun  Tchebysheff's objective function 
   alphacon(k) constraints for ALPHA; 
 
   tch_objfun..   ALPHA + rho*sum(k, (zideal(k)-Z(k))/(zideal(k)-zdefn(k)))=e=TCH_OBJ; 
   alphacon(k).. ALPHA =g= wk(k)*(zideal(k)-Z(k))/(zideal(k)-zdefn(k)); 
 
Model augmtch   / eur_50, tch_objfun, alphacon / ; 
option seed=5780; 
 
*--------------------------------------------------------------------------- 
wk(K)=(wmax(K)+wmin(K))/2; 
solve augmtch using mip minimizing TCH_OBJ; 
*put '    *        Zobj      Return         MAD         DIV        BETA         LIQ  trans_cost    Stck/Portf   '; 
put '    *        Zobj      Return         MAD         DIV  Stck/Portf   '; 
loop(I, put fx I.tl:6); put fx /; 
put '    *'; 
put fx ALPHA.L:12:3 ; 
loop(k, put fx (Z.l(k)):12:4); 
*put fx PORTFBETA.L:12:4, PORTFLIQ.L:12:4 ; 
*    put fx DZ.l('1'):12:4 ; 
*transact=  SUM(MRK,SUM(MRKI(MRK,I),(WB.L(I)*bcom(MRK)+WS.L(I)*scom(MRK)))); 
*put transact:12:4,'  '; 
numstock=sum(I,B.L(I)); 
put numstock:12:0,'  '; 
loop(I, put W.l(I):6:3); 
loop (K, put wk(K):8:4); 
 
*    loop(j,if (B.L(J)=1, put ord(J):4:0, ' ',X.l(J):6:3)); 
put /; 

put 'Monte Carlo iterations' /; 
*$ontext 
totiter=0 ; 
for(iter=1 to MCiter, 
* random generation of project scores from uniform distribution 
    okflag=0; 
    while (okflag=0, 
          loop(K$(ord(K)<card(K)), wk(K)=uniform(wmin(K),wmax(K))); 
          totiter=totiter+1; 
          wk(K)$(ord(K)=card(K))=1 - sum(KP$(ord(KP)<card(KP)),wk(KP)); 
          if (sum(K$(ord(K)=card(K)), wk(K))<=sum(K$(ord(K)=card(K)),wmax(K)) and 
sum(K$(ord(K)=card(K)),wk(K))>=sum(K$(ord(K)=card(K)),wmin(K)), okflag=1) 
          ); 
    solve augmtch using mip minimizing TCH_OBJ; 
    put iter:5:0; 
    put fx ALPHA.L:12:3 ; 
    loop(k, put fx (Z.l(k)):12:4); 
*    put fx PORTFBETA.L:12:4, PORTFLIQ.L:12:4 ; 
*    put fx DZ.l('1'):12:4 ; 
*    transact=  SUM(MRK,SUM(MRKI(MRK,I),(WB.L(I)*bcom(MRK)+WS.L(I)*scom(MRK)))); 
*    put transact:12:4,'  '; 
    numstock=sum(I,B.L(I)); 
    put numstock:12:0,'  '; 
    loop(I, put W.l(I):6:3); 
    loop (K, put wk(K):8:4); 
 
*    loop(j,if (B.L(J)=1, put ord(J):4:0, ' ',X.l(J):6:3)); 
    put /; 
    ); 
*$offtext 
 
finish=jnow; 
elapsed_time=(finish-start)*86400; 
put fx 'Elapsed time: ',elapsed_time:12:2, ' seconds' / ; 
put fx 'Monte Carlo total iterations: ', totiter:8:0 /; 
putclose fx ; 


